

Towards Certified Meta-Programming with Typed Template-Coq

Abhishek Anand, Simon Boulrier, Cyril Cohen, Matthieu Sozeau, Nicolas Tabareau

► **To cite this version:**

Abhishek Anand, Simon Boulrier, Cyril Cohen, Matthieu Sozeau, Nicolas Tabareau. Towards Certified Meta-Programming with Typed Template-Coq. ITP 2018 - 9th Conference on Interactive Theorem Proving, Jul 2018, Oxford, United Kingdom. pp.20-39, 10.1007/978-3-319-94821-8_2 . hal-01809681

HAL Id: hal-01809681

<https://hal.archives-ouvertes.fr/hal-01809681>

Submitted on 7 Jun 2018

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

Towards Certified Meta-Programming with Typed Template-Coq

Abhishek Anand¹, Simon Boulier², Cyril Cohen³,
Matthieu Sozeau⁴ and Nicolas Tabareau²

¹ Cornell University, Ithaca, NY U.S.A.

² Gallinette Project-Team, Inria Nantes, France

³ Université Côte d'Azur, Inria, France

⁴ Pi.R2 Project-Team, Inria Paris and IRIF, France

Abstract. `TEMPLATE-COQ`⁵ is a plugin for `COQ`, originally implemented by Malecha [18], which provides a reifier for `COQ` terms and global declarations, as represented in the `COQ` kernel, as well as a denotation command. Initially, it was developed for the purpose of writing functions on `COQ`'s AST in `GALLINA`. Recently, it was used in the `CERTICOQ` certified compiler project [4], as its front-end language, to derive parametricity properties [3], and to extract `COQ` terms to a CBV λ -calculus [13]. However, the syntax lacked semantics, be it typing semantics or operational semantics, which should reflect, as formal specifications in `COQ`, the semantics of `COQ`'s type theory itself. The tool was also rather bare bones, providing only rudimentary quoting and unquoting commands. We generalize it to handle the entire Calculus of Inductive Constructions (CIC), as implemented by `COQ`, including the kernel's declaration structures for definitions and inductives, and implement a monad for general manipulation of `COQ`'s logical environment. We demonstrate how this setup allows `COQ` users to define many kinds of general purpose plugins, whose correctness can be readily proved in the system itself, and that can be run efficiently after extraction. We give a few examples of implemented plugins, including a parametricity translation. We also advocate the use of `TEMPLATE-COQ` as a foundation for higher-level tools.

1 Introduction

Meta-programming is the art of writing programs (in a *meta-language*) that produce or manipulate programs (written in an *object language*). In the setting of dependent type theory, the expressivity of the language permits to consider the case where the meta and object languages are actually the same, *accounting for well-typedness*. This idea has been pursued in the work on inductive-recursive (IR) and quotient inductive-inductive types (QIIT) in Agda to reflect a syntactic model of a dependently-typed language within another one [9,2]. These term encodings include type-correctness internally by considering only well-typed terms of the syntax, i.e. derivations. However, the use of IR or QIITs complicates considerably

⁵ <https://template-coq.github.io/template-coq>

the meta-theory of the meta-language which makes it difficult to coincide with the object language represented by an inductive type. More problematically in practice, the concision and encapsulation of the syntactic encoding has the drawback that it is very difficult to use because any function from the syntax can be built only at the price of a proof that it respects typing, conversion or any other features described by the intrinsically typed syntax right away.

Other works have taken advantage of the power of dependent types to do meta-programming in a more progressive manner, by first defining the syntax of terms and types; and then defining out of it the notions of reduction, conversion and typing derivation [11,26] (the introduction of [11] provides a comprehensive review of related work in this area). This can be seen as a type-theoretic version of the functional programming language designs such as `TEMPLATE HASKELL` [22] or `METAML` [24]. This is also the approach taken by Malecha in his thesis [18], where he defined `TEMPLATE-COQ`, a plugin which defines a correspondence—using quoting and unquoting functions—between `COQ` kernel terms and inhabitants of an inductive type representing internally the syntax of the calculus of inductive constructions (CIC), as implemented in `COQ`. It becomes thus possible to define programs in `COQ` that manipulate the representation of `COQ` terms and reify them as functions on `COQ` terms. Recently, its use was extended for the needs of the `CERTICOQ` certified compiler project [4], which uses it as its front-end language. It was also used by Anand and Morissett [3] to formalize a modified parametricity translation, and to extract `COQ` terms to a `CBV` λ -calculus [13]. All of these translations however lacked any means to talk about the semantics of the reified programs, only syntax was provided by `TEMPLATE-COQ`. This is an issue for `CERTICOQ` for example where both a non-deterministic small step semantics and a deterministic call-by-value big step semantics for CIC terms had to be defined and preserved by the compiler, without an “official” specification to refer to.

This paper proposes to remedy this situation and provides a formal semantics of `COQ`’s implemented type theory, that can independently be refined and studied. The advantage of having a very concrete untyped description of `COQ` terms (as opposed to IR or QIITs definitions) together with an explicit type checker is that the extracted type-checking algorithm gives rise to an OCAML program that can directly be used to type-check `COQ` kernel terms. This opens a way to a concrete solution to bootstrap `COQ` by implementing the `COQ` kernel in `COQ`. However, a complete reification of CIC terms and a definition of the checker are not enough to provide a meta-programming framework in which `COQ` plugins could be implemented. One needs to get access to `COQ` logical environments. This is achieved using a monad that reifies `COQ` general commands, such as lookups and declarations of constants and inductive types.

As far as we know this is the only reflection framework in a dependently-typed language allowing such manipulations of terms and datatypes, thanks to the relatively concise representation of terms and inductive families in CIC. Compared to the `METACOQ` project [27], `LEAN`’s tactic monad [12], or `AGDA`’s reflection framework [26], our ultimate goal is not to interface with `COQ`’s unification

```

Inductive term : Set :=
| tRel      : ℕ → term
| tVar      : ident → term
| tEvar     : ℕ → list term → term
| tSort     : universe → term
| tCast     : term → cast_kind → term → term
| tProd     : name → term → term → term
| tLambda   : name → term → term → term
| tLetIn    : name → term → term → term → term
| tApp      : term → list term → term
| tConst    : kername → universe_instance → term
| tInd      : inductive → universe_instance → term
| tConstruct: inductive → ℕ → universe_instance → term
| tCase     : inductive * ℕ → term → term → list (ℕ * term) → term
| tProj     : projection → term → term
| tFix      : mfixpoint term → ℕ → term
| tCoFix    : mfixpoint term → ℕ → term.

```

Fig. 1. Representation of the syntax in TEMPLATE-COQ

and type-checking algorithms, but to provide a self-hosted, bootstrappable and verifiable implementation of these algorithms. On one hand, this opens the possibility to verify the kernel’s implementation, a problem tackled by BARRAS [6] using set-theoretic models. On the other hand we also advocate for the use of TEMPLATE-COQ as a foundation on which higher-level tools can be built: meta-programs implementing translations, boilerplate-generating tools, domain-specific proof languages, or even general purpose tactic languages.

Plan of the paper. In Section 2, we present the complete reification of COQ terms, covering the entire CIC and define in Section 3 the type-checking algorithm of COQ reified terms in Coq. In Section 4, we show the definition of a monad for general manipulation of COQ’s logical environment and use it to define plugins for various translations from Coq to Coq (Section 5). Finally, we discuss related and future work in Section 6.

2 Reification of Coq Terms

Reification of syntax. The central piece of TEMPLATE-COQ is the inductive type `term` which represents the syntax of COQ terms, as defined in Fig. 1. This inductive follows directly the `constr` datatype of COQ terms in the OCAML code of COQ, except for the use of OCAML’s native arrays and strings; an upcoming extension of COQ [5] with such features should solve this mismatch.

Constructor `tRel` represents variables bound by abstractions (introduced by `tLambda`), dependent products (introduced by `tProd`) and local definitions (introduced by `tLetIn`), the natural number is a De Bruijn index. The `name` is a printing annotation.

Sorts are represented with `tSort`, which takes a `universe` as argument. A universe is the supremum of a (non-empty) list of level expressions, and a level is either `Prop`, `Set`, a global level or a De Bruijn polymorphic level variable.

`Inductive level := lProp | lSet | Level (_ : string) | Var (_ : ℕ).`

`Definition universe := list (level * bool). (* level+1 if true *)`

The application (introduced by `tApp`) is n-ary. The `tConst`, `tInd` and `tConstruct` constructors represent references to constants (definitions or axioms), inductives, or constructors of an inductive type. The `universe_instances` are non-empty only for polymorphic constants. Finally, `tCase` represents pattern-matchings, `tProj` primitive projections, `tFix` fixpoints and `tCoFix` cofixpoints.

Quoting and unquoting of terms. TEMPLATE-COQ provides a lifting from concrete syntax to reified syntax (quoting) and the converse (unquoting). It can reify and reflect all kernel COQ terms.

The command `Quote Definition` reifies the syntax of a term. For instance,

`Quote Definition f := (fun x => x + 0).`

generates the term `f` defined as

```
f = tLambda (nNamed "x") (tInd {| inductive_mind := "Coq.Init.Datatypes.
  nat"; inductive_ind := 0 |} []) (tApp (tConst "Coq.Init.Nat.add" [])
  [tRel 0; tConstruct {| inductive_mind := "Coq.Init.Datatypes.nat";
    inductive_ind := 0 |} 0 []]) : term
```

On the converse, the command `Make Definition` constructs a term from its syntax. This example below defines `zero` to be 0 of type `N`.

`Make Definition zero := tConstruct (mkInd "Coq.Init.Datatypes.nat" 0)
 0 [].`

where `mkInd n k` is the k^{th} inductive of the mutual block of the name `n`.

Reification of environment. In COQ, the meaning of a term is relative to an environment, which must be reified as well. Environments consist of three parts: (i) a graph of universes (ii) declarations of definitions, axioms and inductives (iii) a local context registering types of De Bruijn indexes.

As we have seen in the syntax of terms, universe levels are not given explicitly in COQ. Instead, level variables are introduced and constraints between them are registered in a *graph of universes*. This is the way typical ambiguity is implemented in COQ. A constraint is given by two levels and a `constraint_type` (`Lt`, `Le` or `Eq`):

`Definition univ_constraint := Level * constraint_type * Level.`

Then the graph is given by a set of level variables and one of constraints. Sets, coming from the COQ standard library, are implemented using lists without duplicates. `LevelSet.t` means the type `t` of the module `LevelSet`.

`Definition uGraph := LevelSet.t * ConstraintSet.t.`

Functions to query the graph are provided, for the moment they rely on a naive implementation of the Bellman-Ford algorithm. `check_leq u1 u2` checks if the graph enforces $u1 \leq u2$ and `no_universe_inconsistency` checks that the graph has no negative cycle.

Constant and inductive declarations are grouped together, properly ordered according to dependencies, in a global context (`global_ctx`), which is a list of global declarations (`global_decl`).

```
Inductive global_decl :=
| ConstantDecl : ident → constant_decl → global_decl
| InductiveDecl : ident → minductive_decl → global_decl.
```

Definitions and axioms just associate a name to a universe context, and two terms for the optional body and type. Inductives are more involved:

```
(* Declaration of one inductive type *)
Record inductive_body := { ind_name : ident;
  ind_type : term; (* closed arity *)
  ind_kelim : list sort_family; (* allowed elimination sorts *)
  (* names, types, number of arguments of constructors *)
  ind_ctors : list (ident * term * nat);
  ind_projs : list (ident * term) (* names and types of projections *)}.

(* Declaration of a block of mutual inductive types *)
Record minductive_decl := { ind_npars : nat; (* number of parameters *)
  ind_bodies : list inductive_body; (* inductives of the mutual block *)
  ind_universes : universe_context (* universe constraints *) }.
```

In COQ internals, there are in fact two ways of representing a declaration: either as a “declaration” or as an “entry”. The kernel takes entries as input, type-check them and elaborate them to declarations. In `TEMPLATE-COQ`, we provide both, and provide an erasing function `mind_decl_to_entry` from declarations to entries for inductive types.

Finally, local contexts are just list of local declarations: a type for lambda bindings and a type and a body for let bindings.

Quoting and unquoting the environment `TEMPLATE-COQ` provides the command `Quote Recursively Definition` to quote an environment. This command crawls the environment and quote all declarations needed to typecheck a given term.

The other way, the commands `Make Inductive` allows declaring an inductive type from its entry. For instance the following redefines a copy of `ℕ`:

```
Make Inductive (mind_decl_to_entry
  { | ind_npars := 0; ind_universes := [];
    ind_bodies := [{
      ind_name := "nat";
      ind_type := tSort [(lSet, false)];
      ind_kelim := [InProp; InSet; InType];
      ind_ctors := [("0", tRel 0, 0);
        ("S", tProd nAnon (tRel 0) (tRel 1), 1)];
      ind_projs := [] |}] |} ).
```

```

Inductive cumul (Σ : global_ctx) (Γ : context) : term → term → Prop
:= | cumul_refl t u : leq_term (snd Σ) t u = true → Σ ; Γ ⊢ t ≤ u
| cumul_red_l t u v : red1 Σ Γ t v → Σ ; Γ ⊢ v ≤ u → Σ ; Γ ⊢ t ≤ u
| cumul_red_r t u v : Σ ; Γ ⊢ t ≤ v → red1 Σ Γ u v → Σ ; Γ ⊢ t ≤ u
where " Σ ; Γ ⊢ t ≤ u " := (cumul Σ Γ t u).

Inductive typing (Σ : global_ctx) (Γ : context) : term → term → Set
:= | type_Rel n : ∀ (H : n < List.length Γ),
    Σ ; Γ ⊢ tRel n : lift₀ (S n) (safe_nth Γ (exist _ n H)).(decl_type)
| type_Sort (l : level) :
    Σ ; Γ ⊢ tSort (Universe.make l) : tSort (Universe.super l)
| type_Prod n t b s1 s2 :
    Σ ; Γ ⊢ t : tSort s1 → Σ ; Γ , vass n t ⊢ b : tSort s2 →
    Σ ; Γ ⊢ tProd n t b : tSort (max_universe s1 s2)
| type_App t l t_ty t' :
    Σ ; Γ ⊢ t : t_ty → typing_spine Σ Γ t_ty l t' →
    Σ ; Γ ⊢ tApp t l : t'
| ...
where " Σ ; Γ ⊢ t : T " := (typing Σ Γ t T)

with typing_spine Σ Γ : term → list term → term → Prop :=
| type_spine_nil ty : typing_spine Σ Γ ty [] ty
| type_spine_const hd t1 na A B B' T :
    Σ ; Γ ⊢ T ≤ tProd na A B → Σ ; Γ ⊢ hd : A →
    typing_spine Σ Γ (subst₀ hd B) t1 B' →
    typing_spine Σ Γ T (cons hd t1) B'

```

Fig. 2. Typing judgment for terms, excerpt

More examples of use of quoting/unquoting commands can be found in the file `test-suite/demo.v`.

3 Type Checking Coq in Coq

In Fig. 2, we present (an excerpt of) the specification of the typing judgment of the kernel of COQ using the inductive type `typing`. It represents all the typing rules of COQ⁶. This includes the basic dependent lambda calculus with lets, global references to inductives and constants, the `match` construct and primitive projections. Universe polymorphic definitions and the well-formedness judgment for global declarations are dealt with as well.

The only ingredients missing are the guard check for fixpoint and productivity of cofixpoints and the positivity condition of mutual (co-) inductive types. They are work-in-progress.

⁶ We do not treat metavariables which are absent from kernel terms and require a separate environment for their declarations.

The typing judgment `typing` is mutually defined with `typing_spine` to account for n-ary applications. Untyped reduction `red1` and cumulativity `cumul` can be defined separately.

Implementation. To test this specification, we have implemented the basic algorithms for type-checking in COQ, that is, we implement type inference: given a context and a term, output its type or produce a type error. All the rules of type inference are straightforward except for cumulativity. The cumulativity test is implemented by comparing head normal forms for a fast-path failure and potentially calling itself recursively, unfolding definitions at the head in COQ's kernel in case the heads are equal. We implemented weak-head reduction by mimicking COQ's kernel implementation, which is based on an abstract machine inspired by the KAM. COQ's machine optionally implements a variant of lazy, memoizing evaluation (which can have mixed results, see COQ's [PR #555](#) for example), that feature has not been implemented yet.

The main difference with the OCAML implementation is that all of the functions are required to be shown terminating in COQ. One possibility could be to prove the termination of type-checking separately but this amounts to prove in particular the normalization of CIC which is a complex task. Instead, we simply add a fuel parameter to make them syntactically recursive and make `OutOfFuel` a type error, *i.e.*, we are working in a variant of the option monad.

Bootstrapping it. We can extract this checker to OCAML and reuse the setup described in Section 2 to connect it with the reifier and easily derive a (partially verified) alternative checker for COQ's `.vo` object files. Our plugin provides a new command `Template Check` for typechecking definitions using the alternative checker, that can be used as follows:

```
Require Import Template.TemplateCoqChecker List. Import ListNotations.
Definition foo := List.map (fun x => x + 3) [0; 1].
Template Check foo.
```

Our initial tests indicate that its running time is comparable to the `coqchk` checker of COQ, as expected.

4 Reification of Coq Commands

COQ plugins need to interact with the environment, for example by repeatedly looking up definitions by name, declaring new constants using fresh names, or performing computations. It is desirable to allow such programs to be written in COQ (`GALLINA`) because of the two following advantages. Plugin-writers no longer need to understand the OCAML implementation of COQ and plugins are no longer sensitive to changes made in the OCAML implementation. Also, when plugins implementing syntactic models are proven correct in COQ, they provide a mechanism to add axioms to COQ without compromising consistency (§5.3).

In general, interactions with the environment have side effects, *e.g.* the declaration of new constants, which must be described in COQ's pure setting. To

overcome this difficulty, we use the standard “free” monadic setting to represent the operations involved in interacting with the environment, as done for instance in Mtac [27].

```

Inductive TemplateMonad : Type → Type :=
(* Monadic operations *)
| tmReturn : ∀ {A}, A → TemplateMonad A
| tmBind   : ∀ {A B},
    TemplateMonad A → (A → TemplateMonad B) → TemplateMonad B

(* General operations *)
| tmPrint   : ∀ {A}, A → TemplateMonad unit
| tmFail    : ∀ {A}, string → TemplateMonad A
| tmEval    : reductionStrategy → ∀ {A}, A → TemplateMonad A
| tmDefinition : ident → ∀ {A}, A → TemplateMonad A
| tmAxiom   : ident → ∀ A, TemplateMonad A
| tmLemma   : ident → ∀ A, TemplateMonad A
| tmFreshName : ident → TemplateMonad ident
| tmAbout   : ident → TemplateMonad (option global_reference)
| tmCurrentModPath : unit → TemplateMonad string

(* Quoting and unquoting operations *)
| tmQuote      : ∀ {A}, A → TemplateMonad term
| tmQuoteRec   : ∀ {A}, A → TemplateMonad program
| tmQuoteInductive : kername → TemplateMonad mutual_inductive_entry
| tmQuoteConstant : kername → bool → TemplateMonad constant_entry
| tmMkDefinition : ident → term → TemplateMonad unit
| tmMkInductive : mutual_inductive_entry → TemplateMonad unit
| tmUnquote    : term → TemplateMonad {A : Type & A}.
| tmUnquoteTyped : ∀ A, term → TemplateMonad A

```

Fig. 3. The monad of commands

`TemplateMonad` is an inductive family (Fig. 3) such that `TemplateMonad A` represents a program which will finally output a term of type `A`. There are special constructors `tmReturn` and `tmBind` to provide (freely) the basic monadic operations. We use the monadic syntactic sugar `x ← t ; u` for `tmBind t (fun x ⇒ u)`.

The other operations of the monad can be classified in two categories: the traditional COQ operations (`tmDefinition` to declare a new definition, ...) and the quoting and unquoting operations to move between COQ term and their syntax or to work directly on the syntax (`tmMkInductive` to declare a new inductive from its syntax for instance). An overview is given in Table 1.

A program `prog` of type `TemplateMonad A` can be executed with the command `Run TemplateProgram prog`. This command is thus an interpreter for `TemplateMonad` programs, implemented in OCAML as a traditional COQ plugin. The term pro-

Vernacular command	Reified command with its arguments	Description
Eval	<code>tmEval red t</code>	Returns the evaluation of <code>t</code> following the evaluation strategy <code>red</code> (<code>cbv</code> , <code>cbn</code> , <code>hnf</code> , <code>all</code> or <code>lazy</code>)
Definition	<code>tmDefinition id t</code>	Makes the definition <code>id := t</code> and returns the created constant <code>id</code>
Axiom	<code>tmAxiom id A</code>	Adds the axiom <code>id</code> of type <code>A</code> and returns the created constant <code>id</code>
Lemma	<code>tmLemma id A</code>	Generates an obligation of type <code>A</code> , returns the created constant <code>id</code> after all obligations close
About or Locate	<code>tmAbout id</code>	Returns <code>Some gr</code> if <code>id</code> is a constant in the current environment and <code>gr</code> is the corresponding global reference. Returns <code>None</code> otherwise.
	<code>tmQuote t</code>	Returns the syntax of <code>t</code> (of type <code>term</code>)
	<code>tmQuoteRec t</code>	Returns the syntax of <code>t</code> and all the declarations on which it depends
	<code>tmQuoteInductive kn</code>	Returns the declaration of the inductive <code>kn</code>
	<code>tmQuoteConstant kn b</code>	Returns the declaration of the constant <code>kn</code> , if <code>b</code> is <code>true</code> the implementation bypass opacity to get the body of the constant
Make Definition	<code>tmMkDefinition id tm</code>	Adds the definition <code>id := t</code> where <code>t</code> is denoted by <code>tm</code>
Make Inductive	<code>tmMkInductive d</code>	Declares the inductive denoted by the declaration <code>d</code>
	<code>tmUnquote tm</code>	Returns the pair <code>(A;t)</code> where <code>t</code> is the term whose syntax is <code>tm</code> and <code>A</code> it's type
	<code>tmUnquoteTyped A tm</code>	Returns the term whose syntax is <code>tm</code> and checks that it is indeed of type <code>A</code>

Table 1. Main TEMPLATE-COQ commands

duced by the program is discarded but, and it is the point, a program can have many side effects like declaring a new definition or a new inductive type, printing something,

Let's look at some examples. The following program adds the definitions `foo := 12` and `bar := foo + 1` to the current context.

```
Run TemplateProgram (foo ← tmDefinition "foo" 12 ;
                    tmDefinition "bar" (foo +1)).
```

The program below asks the user to provide an inhabitant of \mathbb{N} (here we provide `3 * 3`) and records it in the lemma `foo` ; prints its normal form ; and records the syntax of its normal form in `foo_nf_syntax` (hence of type `term`).

We use PROGRAM’s obligation mechanism⁷ to ask for missing proofs, running the rest of the program when the user finishes providing it. This enables the implementation of *interactive* plugins.

```
Run TemplateProgram (foo ← tmLemma "foo" N ;
                    nf ← tmEval all foo ;
                    tmPrint "normal form: " ; tmPrint nf ;
                    nf_ ← tmQuote nf ;
                    tmDefinition "foo_nf_syntax" nf_).
Next Obligation. exact (3 * 3). Defined.
```

5 Writing Coq plugins in Coq

The reification of syntax, typing and commands of COQ allow writing a COQ plugin directly inside COQ, without requiring another language like OCAML and an external compilation phase.

In this section, we describe three examples of such plugins: (i) a plugin that adds a constructor to an inductive type, (ii) a re-implementation of LASSON’s parametricity plugin⁸, and (iii) an implementation of a plugin that provides an extension of CIC—using a syntactic translation—in which it is possible to prove the negation of functional extensionality [8].

5.1 A Plugin to Add a Constructor

Our first example is a toy example to show the methodology of writing plugins in TEMPLATE-COQ. Given an inductive type I , we want to declare a new inductive type I' which corresponds to I plus one more constructor.

For instance, let’s say we have a syntax for lambda calculus:

```
Inductive tm : Set :=
  | var : nat → tm | lam : tm → tm | app : tm → tm → tm.
```

And that in some part of our development, we want to consider a variation of tm with a new constructor, *e.g.*, `letin`. Then we declare tm' with the plugin by:

```
Run TemplateProgram
  (add_constructor tm "letin" (fun tm' ⇒ tm' → tm' → tm')).
```

This command has the same effect as declaring the inductive tm' by hand:

```
Inductive tm' : Set :=
  | var' : nat → tm' | lam' : tm' → tm'
  | app' : tm' → tm' → tm' | letin : tm' → tm' → tm'.
```

⁷ In COQ, a proof obligation is a goal which has to be solved to complete a definition. Obligations were introduced by SOZEAU [23] in the PROGRAM mode.

⁸ <https://github.com/parametricity-coq/paramcoq>

but with the benefit that if `tm` is changed, for instance by adding one new constructor, then `tm'` is automatically changed accordingly. We provide other examples in the file `test-suite/add_constructor.v`, e.g. with mutual inductives.

We will see that it is fairly easy to define this plugin using `TEMPLATE-COQ`. The main function is `add_constructor` which takes an inductive type `ind` (whose type is not necessarily `Type` if it is an inductive family), a name `idc` for the new constructor and the type `ctor` of the new constructor, abstracted with respect to the new inductive.

```

Definition add_constructor {A} (ind : A) (idc : ident) {B} (ctor : B)
  : TemplateMonad unit
:= tm ← tmQuote ind ;
  match tm with
| tInd ind0 _ =>
  decl ← tmQuoteInductive (inductive_mind ind0) ;
  ctor ← tmQuote ctor ;
  d' ← tmEval lazy (add_ctor decl ind0 idc ctor) ;
  tmMkInductive d'
| _ => tmFail "The provided term is not an inductive"
end.

```

It works in the following way. First the inductive type `ind` is quoted, the obtained term `tm` is expected to be a `tInd` constructor otherwise the function fails. Then the declaration of this inductive is obtained by calling `tmQuoteInductive`, the constructor is reified too, and an auxiliary function is called to add the constructor to the declaration. After evaluation, the new inductive type is added to the current context with `tmMkInductive`.

It remains to define the `add_ctor` auxiliary function to complete the definition of the plugin. It takes a `minductive_decl` which is the declaration of a block of mutual inductive types and returns a `minductive_decl`.

```

Definition add_ctor (mind : minductive_decl) (ind0 : inductive)
  (idc : ident) (ctor : term) : minductive_decl
:= let i0 := inductive_ind ind0 in
  { | ind_npars := mind.(ind_npars) ;
    ind_bodies := map_i (fun (i : nat) (ind : inductive_body) =>
      { | ind_name := tsl_ident ind.(ind_name) ;
        ind_type := ind.(ind_type) ;
        ind_kelim := ind.(ind_kelim) ;
        ind_ctors :=
          let ctors := map (fun '(id, t, k) => (tsl_ident id, t, k))
            ind.(ind_ctors) in
          if Nat.eqb i i0 then
            let n := length mind.(ind_bodies) in
            let typ := try_remove_n_lambdas n ctor in
            ctors ++ [(idc, typ, 0)]
          else ctors ;
        ind_projs := ind.(ind_projs) |})
    mind.(ind_bodies) |}.

```

$$\begin{array}{c}
[t]_0 = t \\
[x]_1 = x^t \\
[\forall(x : A).B]_1 = \lambda f. \forall(x : [A]_0)(x^t : [A]_1 x). [B]_1 (f x) \\
[\lambda(x : A).t]_1 = \lambda(x : [A]_0)(x^t : [A]_1 x). [t]_1 \\
\llbracket \Gamma, x : A \rrbracket = \llbracket \Gamma \rrbracket, x : [A]_0, x^t : [A]_1 x
\end{array}
\quad \left| \quad
\begin{array}{c}
\Gamma \vdash t : A \\
\hline
\llbracket \Gamma \rrbracket \vdash [t]_0 : [A]_0 \\
\llbracket \Gamma \rrbracket \vdash [t]_1 : [A]_1 [t]_0
\end{array}$$

Fig. 4. Unary parametricity translation and soundness theorem, excerpt (from [7])

The declaration of the block of mutual inductive types is a record. The field `ind_bodies` contains the list of declarations of each inductive of the block. We see that most of the fields of the records are propagated, except for the names which are translated to add some primes and `ind_ctors`, the list of types of constructors, for which, in the case of the relevant inductive (`i0` is its number), the new constructor is added.

5.2 Parametricity Plugin

We now show how `TEMPLATE-COQ` permits to define a parametricity plugin that computes the translation of a term following Reynolds' parametricity [21,25]. We follow the already known approaches of parametricity for dependent type theories [7,15], and provide an alternative to Keller and Lasson's plugin.

The definition in the unary case is described in Fig. 4. The soundness theorem ensures that, for a term t of type A , $[t]_1$ computes a proof of parametricity of $[t]_0$ in the sense that it has type $[A]_1 [t]_0$. The definition of the plugin goes in two steps: first the definition of the translation on the syntax of `term` in `TEMPLATE-COQ` and then the instrumentation to connect it with terms of `COQ` using the `TemplateMonad`. It can be found in the file `translations/tsl_param.v`.

The parametricity translation of Fig. 4 is total and syntax directed, the two components of the translation $[]_0$ and $[]_1$ are implemented by two recursive functions `tsl_param0` and `tsl_param1`.

```

Fixpoint tsl_param0 (n : nat) (t : term) {struct t} : term :=
match t with
| tRel k => if k >= n then (* global variable *) tRel (2*k-n+1)
           else (* local variable *) tRel k
| tProd na A B => tProd na (tsl_param0 n A) (tsl_param0 (n+1) B)
| _ => ...
end.

```

```

Fixpoint tsl_param1 (E : tsl_table) (t : term) : term :=
match t with
| tRel k => tRel (2 * k)

```

```

| tSort s ⇒ tLambda (nNamed "A") (tSort s)
                    (tProd nAnon (tRel 0) (tSort s))
| tProd na A B ⇒
  let AO := tsl_param_0 0 A in let A1 := tsl_param_1 E A in
  let BO := tsl_param_0 1 B in let B1 := tsl_param_1 E B in
  tLambda (nNamed "f") (tProd na AO BO)
    (tProd na (lift_0 1 AO)
              (tProd (tsl_name na) (subst_app (lift_0 2 A1) [tRel 0])
                    (subst_app (lift 1 2 B1) [tApp (tRel 2) [tRel 1] ]))))
| tConst s univs ⇒ match lookup_tsl_table E (ConstRef s) with
  | Some t ⇒ t
  | None ⇒ default_term
end
| _ ⇒ ...
end.

```

On Fig. 4, the translation is presented in a named setting, so the introduction of new variables does not change references to existing ones. That's why, $[]_0$ is the identity. In the De Bruijn setting of `TEMPLATE-COQ`, the translation has to take into account the shift induced by the duplication of the context. Therefore, the implementation `tsl_param_0` of $[]_0$ is not the identity anymore. The argument `n` of `tsl_param_0` represents the De Bruijn level from which the variables have been duplicated. There is no need for such an argument in `tsl_param_1`, the implementation of $[]_1$, because in this function all variables are duplicated.

The parametricity plugin not only has to be defined on terms of CIC but also on additional terms dealing with the global context. In particular, constants are translated using a translation table which records the translations of previously processed constants.

Definition `tsl_table := list (global_reference * term)`.

If a constant is not in the translation table we return a dummy `default_term`, considered as an error (this could also be handled by an option monad).

We have also implemented the translation of inductives and pattern matching. For instance the translation of the equality type `eq` produces the inductive type:

```

Inductive eqt A (At : A → Type) (x : A) (xt : At x)
  : ∀ H, At H → x = H → Prop :=
  | eq_reflt : eqt A At x xt x xt eq_refl.

```

Then $[eq]_1$ is given by `eqt` and $[eq_refl]_1$ by `eq_reflt`.

Given `tsl_param_0` and `tsl_param_1` the translation of the declaration of a block of mutual inductive types is not so hard to get. Indeed, such a declaration mainly consists of the arities of the inductives and the types of constructors ; and the one of the translated inductive are produced by translation of the original ones.

Definition `tsl_mind_decl (E : tsl_table) (kn : kername)
 (mind : minductive_decl) : minductive_decl`.

In a similar manner, we can translate pattern-matching. Note that the plugin does not support fixpoints and cofixpoints for the moment.

Now, it remains to connect this translation defined on reified syntax `term` to terms of COQ. For this, we define the new command `tTranslate` in the `TemplateMonad`.

```
Definition tTranslate (E : tsl_table) (id : ident)
  : TemplateMonad tsl_table.
```

When `id` is a definition, the command recovers the body of `id` (as a `term`) using `tmQuoteConstant` and then translates it and records it in a new definition `idt`. The command returns the translation table `E` extended by `(id, idt)`. In the case `id` is an inductive type or a constructor then the command does basically the same but extends the translation table with both the inductive and the constructors. If `id` is an axiom or not a constant the command fails.

Here is an illustration coming from the work of Lasson [16] on the automatic proofs of (ω -)groupoid laws using parametricity. We show that all function of type $ID := \forall A \ x \ y, \ x = y \rightarrow x = y$ are identity functions. First we need to record the translations of `eq` and `ID` in a term table of type `tsl_table`.

```
Run TemplateProgram (table ← tTranslate [] "eq" ;
  table ← tTranslate table "ID" ;
  tmDefinition "table" table).
```

Then we show that every parametric function on `ID` is pointwise equal to the identity using the predicate `fun y => x = y`.

```
Lemma param_ID (f : ID) : IDt f → ∀ A x y p, f A x y p = p.
```

Proof.

```
  intros H A x y p. destruct p.
  destruct (H A (fun y => x = y) x eq_refl
    x eq_refl eq_refl (eq_reflt _ _)).
  reflexivity.
```

Qed.

Then we define a function $myf := p \mapsto p \cdot p^{-1} \cdot p$ and get its parametricity proof using the plugin.

```
Definition myf : ID := fun A x y p => eq_trans (eq_trans p (eq_sym p)) p.
```

```
Run TemplateProgram (table ← tTranslate table "eq_sym" ;
  table ← tTranslate table "eq_trans" ;
  tTranslate table "myf").
```

It is then possible to deduce automatically that $p \cdot p^{-1} \cdot p = p$ for all $p : x = y$.

```
Definition free_thm_myf : ∀ A x y p, myf A x y p = p := param_ID myf myft.
```

5.3 Intensional Function Plugin

Our last illustration is a plugin that provides an intensional flavour to functions and thus allows negating functional extensionality (`FunExt`). This is a simple example of syntactical translation which enriches the logical power of COQ, in

$$\begin{array}{ll}
[x] := x & [\lambda(x : A). t] := (\lambda(x : [A]). [t], \text{true}) \\
[t \ u] := (\pi_1 [t]) [u] & [\forall(x : A). B] := (\forall(x : [A]). [B]) \times \mathbb{B}
\end{array}$$

Fig. 5. Intensional Function Translation, excerpt (from [8])

the sense that new theorems can be proven (as opposed to the parametricity translation which is conservative over CIC). See [8] for an introduction to syntactical translations and a complete description of the intensional function translation.

Even if the translation is very simple as it just adds a boolean to every function (Fig. 5), this time, it is not fully syntax directed. Indeed the notation for pairs hide some types:

```
[fun (x:A) => t] := pair (forall x:[A]. ?T) bool (fun (x:[A]) => [t]) true
```

and we can not recover the type `?T` from the source term. There is thus a mismatch between the lambdas which are not fully annotated and the pairs which are.⁹

However we can use the type inference algorithm of Section 3 implemented on `TEMPLATE-COQ` terms to recover the missing information.

```
[fun (x:A) => t] := let B := infer Sigma (Gamma, x:[A]) t in
  pair (forall (x:[A]). B) bool (fun (x:[A]) => [t]) true
```

Compared to the parametricity plugin, the translation function has a more complex type as it requires the global and local contexts. However, we can generalize the `tTranslate` command so that it can be used for both the parametricity and the intensional function plugins. The implementation is in the files `translations/translation_utils.v` and `translations/tsl_fun.v`.

Extending COQ using plugins. The intensional translation extends the logical power of COQ as it is possible for instance to negate `FunExt`. In this perspective, we defined a new command:

```
Definition tImplement (Sigma : global_ctx * tsl_table)
  (id : ident) (A : Type)
  : TemplateMonad (global_ctx * tsl_table).
```

which computes the translation `A'` of `A`, then asks the user to inhabit the type `A'` by generating a proof obligation and then safely adds the axiom `id` of type `A` to the current context. By safely, we mean that the correction of the translation ensures that no inconsistencies are introduced.

For instance, here is how to negate `FunExt`. We use for that two pairs `(fun x => x; true)` and `(fun x => x; false)` in the interpretation of functions from `unit` to `unit`, which are extensionally both the identity, but differ intensionally on their boolean.

⁹ Note that there is a similar issue with applications and projections, but which can be circumvented this time using (untyped) primitive projections.

```

Run TemplateProgram (TC ← tTranslate ([],[]) "eq" ;
                    TC ← tTranslate TC "False" ;
                    tImplement TC "notFunext"
  ((∀ A B (f g : A → B), (∀ x:A, f x = g x) → f = g) → False)).
Next Obligation.
  tIntro H. tSpecialize H unit. tSpecialize H unit.
  tSpecialize H (fun x => x; true). tSpecialize H (fun x => x; false).
  tSpecialize H (fun x => eq_reflt _ _; true).
  apply eqt_eq in H; discriminate.
Defined.

```

where `tIntro` and `tSpecialize` are special versions of the corresponding `intro` and `specialize` tactics of COQ to deal with extra booleans appearing in the translated terms. After this command, the axiom `notFunext` belongs to the environment, as if it were added with the `Axiom` command. But as we have inhabited the translation of its type, the correctness of the translation ensures that no inconsistency were introduced.

Note that we could also define another translation, e.g. the setoid translation, in which `FunExt` is inhabited. This is not contradictory as the two translations induce two different logical extensions of COQ, which can not be combined.

6 Related Work and Future Work

Meta-Programming is a whole field of research in the programming languages community, we will not attempt to give a detailed review of related work here. In contrast to most work on meta-programming, we provide a very rough interface to the object language: one can easily build ill-scoped and ill-typed terms in our framework, and staging is basic. However, with typing derivations we provide a way to verify meta-programs and ensure that they do make sense.

The closest cousin of our work is the Typed Syntactic Meta-Programming [11] proposal in AGDA, which provides a well-scoped and well-typed interface to a denotation function, that can be used to implement tactics by reflection. We could also implement such an interface, asking for a proof of well-typedness on top of the `tmUnquoteTyped` primitive of our monad.

Intrinsically typed representations of terms in dependent type-theory is an area of active research. Most solutions are based on extensions of Martin-Löf Intensional Type Theory with inductive-recursive or quotient inductive-inductive types [9,2], therefore extending the meta-theory. Recent work on verifying soundness and completeness of the conversion algorithm of a dependent type theory (with natural numbers, dependent products and a universe) in a type theory with IR types [1] gives us hope that this path can nonetheless be taken to provide the strongest guarantees on our conversion algorithm. The intrinsically-typed syntax used there is quite close to our typing derivations.

Another direction is taken by the `CEuf` certified compiler [19], which restricts itself to a fragment of COQ for which a total denotation function can be defined, in the tradition of definitional interpreters advocated by Chlipala [10]. This setup should be readily accomodated by `TEMPLATE-COQ`.

The translation+plugin technique paves the way for certified translations and the last piece will be to prove correctness of such translations. By correctness we mean computational soundness and typing soundness (see [8]), and both can be stated in `TEMPLATE-COQ`. Anand has made substantial attempts in this direction to prove the computational soundness, in `TEMPLATE-COQ`, of a variant of parametricity providing stronger theorems for free on propositions [3]. This included as a first step a move to named syntax that could be reused in other translations.

Our long term goal is to leverage this technique to extend the logical and computational power of `COQ` using, for instance, the forcing translation [14] or the weaning translation [20].

When performance matters, we can extract the translation to `OCAML` and use it like any ordinary `COQ` plugin. This relies on the correctness of extraction, but in the untyped syntax + typing judgment setting, extraction of translations is almost an identity pretty-printing phase, so we do not lose much confidence. We can also implement a template monad runner in `OCAML` to run the plugins outside `COQ`. Our first experiments show that we could gain a factor 10 for the time needed to compute the translation of a term. Another solution would be to use the certified `CERTICOQ` compiler, once it supports a kind of foreign function interface, to implement the `TemplateMonad` evaluation.

The last direction of extension is to build higher-level tools on top of the syntax: the unification algorithm described in [28] is our first candidate. Once unification is implemented, we can look at even higher-level tools: elaboration from concrete syntax trees, unification hints like canonical structures and type class resolution, domain-specific and general purpose tactic languages. A key inspiration in this regard is the work of Malecha and Bengston [17] which implemented this idea on a restricted fragment of `CIC`.

Acknowledgments

This work is supported by the `CoqHoTT` ERC Grant 64399 and the NSF grants CCF-1407794, CCF-1521602, and CCF-1646417.

References

1. Abel, A., Öhman, J., Vezzosi, A.: [Decidability of conversion for type theory in type theory](#). *PACMPL* 2(POPL), 23:1–23:29 (2018)
2. Altenkirch, T., Kaposi, A.: [Type Theory in Type Theory Using Quotient Inductive Types](#). pp. 18–29. *POPL '16*, ACM, New York, NY, USA (2016)
3. Anand, A., Morrisett, G.: [Revisiting Parametricity: Inductives and Uniformity of Propositions](#). In: *CoqPL'18*. Los Angeles, CA, USA (2018)
4. Anand, A., Appel, A., Morrisett, G., Paraskevopoulou, Z., Pollack, R., Belanger, O.S., Sozeau, M., Weaver, M.: [CertiCoq: A verified compiler for Coq](#). In: *CoqPL*. Paris, France (2017)

5. Armand, M., Grégoire, B., Spiwack, A., Théry, L.: Extending Coq with Imperative Features and Its Application to SAT Verification. In: Kaufmann, M., Paulson, L.C. (eds.) *Interactive Theorem Proving*. pp. 83–98. Springer (2010)
6. Barras, B.: Auto-validation d’un système de preuves avec familles inductives. Thèse de doctorat, Université Paris 7 (Nov 1999)
7. Bernardy, J.P., Jansson, P., Paterson, R.: Proofs for free: Parametricity for dependent types. *Journal of Functional Programming* 22(2), 107–152 (Mar 2012)
8. Boulrier, S., Pédrot, P.M., Tabareau, N.: The Next 700 Syntactical Models of Type Theory. In: *CPP’17*, Paris, France. pp. 182–194. ACM (2017)
9. Chapman, J.: **Type Theory Should Eat Itself**. *Electronic Notes in Theoretical Computer Science* 228, 21 – 36 (2009), proceedings of LFMTP 2008
10. Chlipala, A.: *Certified Programming with Dependent Types*, vol. 20. MIT Press (2011)
11. Devriese, D., Piessens, F.: **Typed Syntactic Meta-programming**. *ICFP ’13*, vol. 48, pp. 73–86. ACM (2013)
12. Ebner, G., Ullrich, S., Roesch, J., Avigad, J., de Moura, L.: A Metaprogramming Framework for Formal Verification pp. 34:1–34:29 (Sep 2017)
13. Forster, Y., Kunze, F.: **Verified Extraction from Coq to a Lambda-Calculus**. In: *Coq Workshop 2016* (2016)
14. Jaber, G., Lewertowski, G., Pédrot, P.M., Sozeau, M., Tabareau, N.: **The Definitional Side of the Forcing**. In: *LICS’16*, New York, NY, USA. pp. 367–376 (2016)
15. Keller, C., Lasson, M.: **Parametricity in an Impredicative Sort**. *CoRR* abs/1209.6336 (2012)
16. Lasson, M.: Canonicity of Weak ω -groupoid Laws Using Parametricity Theory. *Electronic Notes in Theoretical Computer Science* 308, 229–244 (2014)
17. Malecha, G., Bengtson, J.: *ESOP 2016*, chap. Extensible and Efficient Automation Through Reflective Tactics, pp. 532–559. Springer Berlin Heidelberg (2016)
18. Malecha, G.M.: *Extensible Proof Engineering in Intensional Type Theory*. Ph.D. thesis, Harvard University (2014)
19. Mullen, E., Pernsteiner, S., Wilcox, J.R., Tatlock, Z., Grossman, D.: **EuF: minimizing the Coq extraction TCB**. In: *Proceedings of CPP 2018*. pp. 172–185 (2018)
20. Pédrot, P., Tabareau, N.: **An effectful way to eliminate addiction to dependence**. In: *LICS’17*, Reykjavik, Iceland. pp. 1–12 (2017)
21. Reynolds, J.C.: Types, Abstraction and Parametric Polymorphism. In: *IFIP Congress*. pp. 513–523 (1983)
22. Sheard, T., Jones, S.P.: **Template Meta-programming for Haskell**. *SIGPLAN Not.* 37(12), 60–75 (Dec 2002)
23. Sozeau, M.: **Program-ing Finger Trees in Coq**. pp. 13–24. *ICFP ’07*, ACM, New York, NY, USA (2007)
24. Taha, W., Sheard, T.: **Multi-stage Programming with Explicit Annotations**. pp. 203–217. *PEPM ’97*, ACM, New York, NY, USA (1997)
25. Wadler, P.: Theorems for Free! In: *Functional Programming Languages and Computer Architecture*. pp. 347–359. ACM Press (1989)
26. Van der Walt, P., Swierstra, W.: Engineering Proof by Reflection in Agda. In: *Implementation and Application of Functional Languages*. Springer (2013)
27. Ziliani, B., Dreyer, D., Krishnaswami, N.R., Nanevski, A., Vafeiadis, V.: **Mtac: A Monad for Typed Tactic Programming in Coq**. *Journal of Functional Programming* 25 (2015)
28. Ziliani, B., Sozeau, M.: **A Comprehensible Guide to a New Unifier for CIC Including Universe Polymorphism and Overloading**. *Journal of Functional Programming* 27, e10 (2017)